
Subject: Gini - Income Inequality
Posted by [ahmed89o](#) on Fri, 15 Aug 2014 22:12:18 GMT
[View Forum Message](#) <> [Reply to Message](#)

Dear DHS team,
I have noted that in your recent reports you start to provide Gini by City. I am assuming you have built using wealth index. I would like to kindly ask you to share with your do file for stata on how to develop Gini by city so I can run for older surveys. All the best,

Subject: Re: Gini - Income Inequality
Posted by [Trevor-DHS](#) on Mon, 18 Aug 2014 03:54:46 GMT
[View Forum Message](#) <> [Reply to Message](#)

Dear Ahmed89o

Unfortunately we do not have a Stata do file for calculating the Gini coefficient. We use CPro for producing the tabulations for the DHS reports, and the code is written in CPro.

You can find an explanation of the calculation of the Gini coefficient in the DHS6 tabulation plan (http://www.dhsprogram.com/pubs/pdf/DHSM6/Final_Report_Tab_Plan_05June2013_DHSM6.pdf) starting on page 12.

Subject: Re: Gini - Income Inequality
Posted by [ahmed89o](#) on Mon, 18 Aug 2014 08:38:41 GMT
[View Forum Message](#) <> [Reply to Message](#)

Dear DHS team,
I have looked at the report and it explains the idea of Gini. But I want to be informed how did you deal with negative values in wealth index. What kind of transformation did you use? Can you please share the code of cpro and I will try to use it? I will try to download it and replicate your estimates.

Subject: Re: Gini - Income Inequality
Posted by [ahmed89o](#) on Sun, 07 Sep 2014 19:44:33 GMT
[View Forum Message](#) <> [Reply to Message](#)

Dear Dhs team,
I have not answer regarding my second post. I would like how you death with negative values on the wealth index to build gini

Subject: Re: Gini - Income Inequality
Posted by [Trevor-DHS](#) on Mon, 08 Sep 2014 16:49:23 GMT

Here is some code for calculating the Gini coefficient. To run this for subgroups, drop the cases not needed - see example:

```
use "BDHR61FL.DTA", clear
```

```
* Use a selection here to run this for a specific subgroup
```

```
* e.g. for urban
```

```
* drop if hv025!=1
```

```
* Summarize and get the minimum and the maximum
```

```
quietly summ hv271
```

```
local w_min = r(min)
```

```
local w_max = r(max)
```

```
* Calculating the range
```

```
local w_range = `w_max' - `w_min'
```

```
* Create 100 groups
```

```
gen w_group = int( (hv271-`w_min') / (`w_range'/(100-1)) ) + 1
```

```
* Transformed wealth score - 0 based
```

```
gen wscore_trans = hv271 - `w_min'
```

```
* Summarize by the 100 groups
```

```
collapse (sum) pop=hv012 ws=wscore_trans [pw=hv005/1000000], by(w_group)
```

```
* Accumulate population and wealth scores across groups
```

```
gen pop_accum = pop
```

```
replace pop_accum = pop_accum[_n-1] + pop if _n>1
```

```
gen wdx_accum = ws
```

```
replace wdx_accum = wdx_accum[_n-1] + ws if _n>1
```

```
* Sum total population and total wealth scores
```

```
quietly summ pop
```

```
local pop_tot = r(sum)
```

```
quietly summ ws
```

```
local wdx_tot = r(sum)
```

```
* Calculate proportion in each group for population and wealth
```

```
gen pop_prop = pop_accum / `pop_tot'
```

```
gen wdx_prop = wdx_accum / `wdx_tot'
```

```
* Calculate Gini coefficient elements
```

```
gen gini = (pop_prop - pop_prop[_n-1]) * (wdx_prop + wdx_prop[_n-1]) if _n>1
```

```
* Gini coefficient is 1 - sum of elements, multiplied by 100 to be a percentage
```

```
quietly summ gini
```

```
local gini_coeff 100 * (1 - r(sum))
```

```
di "Gini coefficient: " `gini_coeff'
```

Subject: Re: Gini - Income Inequality
Posted by [ahmed89o](#) on Tue, 09 Sep 2014 09:12:27 GMT
[View Forum Message](#) <> [Reply to Message](#)

Thanks a million Trevor for this. In case I want to get gini by city or governorate. It seems to me I can simply run the do file on each city or governorate or subsample separately, right?

Subject: Re: Gini - Income Inequality
Posted by [Trevor-DHS](#) on Tue, 09 Sep 2014 14:29:50 GMT
[View Forum Message](#) <> [Reply to Message](#)

You can use the drop command in the 5th line. So, if you wanted to produce an estimate just for urban areas, drop those areas that are not urban with
drop hv025!=1
If you wanted just the second governorate, for example, use
drop hv024!=2

I could probably make the code work to produce the Gini coefficient by background variables, but it would take rewriting the code in a different way - not using summarize and not using the locals - but it seems easy enough just to add in the drop to select the cases you want.

Subject: Re: Gini - Income Inequality
Posted by [schoumaker](#) on Fri, 26 Feb 2016 12:26:12 GMT
[View Forum Message](#) <> [Reply to Message](#)

Thanks a lot for providing this code. For information, the ineqdeco Stata command can also be used to compute the Gini index. This gives the same results as Trevor's syntax.
Best,
Bruno

Subject: Re: Gini - Income Inequality
Posted by [ttuti](#) on Fri, 15 Apr 2016 16:50:31 GMT
[View Forum Message](#) <> [Reply to Message](#)

Is it possible to get the formula to generate the GINI coefficient in a software independent kind of way? Having a difficult time translating this Stata code into R equivalent. I am trying to calculate the GINI coefficient using the entire country sample and in RStudio (R). Any help is welcomed.

Subject: Re: Gini - Income Inequality
Posted by [Trevor-DHS](#) on Mon, 18 Apr 2016 18:12:39 GMT
[View Forum Message](#) <> [Reply to Message](#)

See the DHS tabulation plan, pages 12-13 for an explanation of the Gini index and its calculation.

Subject: Re: Gini - Income Inequality

Posted by [r01efa16](#) on Thu, 01 Jun 2017 12:49:39 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hello

This file is no longer available

Please I need the code and steps to run the command in STATA

Thanks in advance

Subject: Re: Gini - Income Inequality

Posted by [Trevor-DHS](#) on Thu, 01 Jun 2017 16:26:37 GMT

[View Forum Message](#) <> [Reply to Message](#)

For the file you say is not available, which file is that? If you mean the DHS tabulation plan, pages 12-13, the link still works fine for me - can you test the link again.

The Stata code is already shown in this thread, so I'm not sure what else you need.
